

Logic Programming; Dynamic Types
Section and Practice Problems

Section 12

1 Logic Programming

To try playing around with Prolog, go to <http://www.swi-prolog.org/>. You will be able to use Prolog online at <https://swish.swi-prolog.org/>.

To try playing around with Datalog, you can go to either <http://abcdatalog.seas.harvard.edu/> to download a Java-based Datalog implementation, or you can go to <https://datalog.db.in.tum.de/> to use Datalog online.

Although you can use the tools above to get the answers to the section problems below very easily, work out the answers by hand (to make sure you understand the semantics of Prolog and Datalog), and then you can check your answers by using the tools to execute the programs.

- (a) Consider the following Prolog program (where $[]$ is a constant representing the empty list, $[t]$ is shorthand for $\text{cons}(t, [])$ and $[t_1, t_2|t_3]$ is shorthand for $\text{cons}(t_1, \text{cons}(t_2, t_3))$).

```
foo([], []).
foo([X], [X]).
foo([X, Y|S], [Y, X|T]) :- foo(S, T).
```

For each of the following queries, compute the substitutions that Prolog will generate, if any. (Note that there is a difference between an empty substitution, and no substitution.) If the query evaluation will not terminate, explain why.

- $\text{foo}([a, b], X)$.
- $\text{foo}([a, b, c], X)$.
- $\text{foo}([a, b], [a, b])$
- $\text{foo}(X, [a])$
- $\text{foo}(X, Y)$.

Answer: Intuitively, $\text{foo}(S, T)$ holds for two lists S and T if they are the same length, and for all i , the $2i$ th and $2i + 1$ th elements of S are equal, respectively, to the $2i + 1$ th and $2i$ th elements of T .

- $\text{foo}([a, b], X)$
 $X = [b, a]$
- $\text{foo}([a, b, c], X)$
 $X = [b, a, c]$
- $\text{foo}([a, b], [a, b])$
No substitutions returned
- $\text{foo}(X, [a])$
 $X = [a]$

- `foo(X, Y).`
`X = [], Y = []`
`X = [A, B], Y = [B, A]`
`X = [A, B, C], Y = [B, A, C]`
`X = [A, B, C, D], Y = [B, A, D, C]`
`X = [A, B, C, D, E], Y = [B, A, D, C, E]`
`X = [A, B, C, D, E, F], Y = [B, A, D, C, F, E]`
...
The evaluation of the query never terminates.

(b) Consider the following Datalog program.

```

bar(a, b, c).
bar(X, Y, Z) :- bar(Y, X, Z).
bar(X, Y, Z) :- bar(Z, Y, X), quux(X, Z).
quux(b, c).
quux(c, d).
quux(X, Y) :- quux(Y, X).
quux(X, Z) :- quux(X, Y), quux(Y, Z).

```

Find all solutions to the query `bar(X, Y, Z).`

Answer: We start by the set of facts that are known, S_0 , and then given S_i we produce S_{i+1} by unifying the horn clauses with the known facts to derive new facts, and repeat until we reach a fixed point.

```

S0 = {bar(a, b, c)., quux(b, c)., quux(c, d).}
S1 = {bar(a, b, c)., quux(b, c)., quux(c, d)., bar(b, a, c)., quux(c, b)., quux(d, c)., quux(b, d).}
S2 = {bar(a, b, c)., quux(b, c)., quux(c, d)., bar(b, a, c)., quux(c, b)., quux(d, c)., quux(b, d).,
      bar(c, a, b)., quux(d, b)., quux(b, b).quux(c, c).}
S3 = {bar(a, b, c)., quux(b, c)., quux(c, d)., bar(b, a, c)., quux(c, b)., quux(d, c)., quux(b, d).,
      bar(c, a, b)., quux(d, b)., quux(b, b).quux(c, c)., bar(a, c, b).}
S4 = {bar(a, b, c)., quux(b, c)., quux(c, d)., bar(b, a, c)., quux(c, b)., quux(d, c)., quux(b, d).,
      bar(c, a, b)., quux(d, b)., quux(b, b).quux(c, c)., bar(a, c, b).}

```

Since S_3 and S_4 are the same (i.e., applying the rules to S_3 doesn't derive any new facts) we have a fixed point. So all solutions to the query `bar(X, Y, Z)?` are:


```

bar(a, b, c).
bar(b, a, c).
bar(c, a, b).
bar(a, c, b).

```

(c) Suppose that we represent a directed graph using the predicates `edge(X, Y)` to indicate that there is

an edge from node X to node Y . For example, the following graph is represented by the following facts:



```
node(a).
node(b).
node(c).
node(d).
edge(a, b).
edge(b, c).
edge(c, d).
edge(d, b).
```

- (i) Write a Datalog program that computes $\text{reachable}(X, Y)$, where $\text{reachable}(X, Y)$ holds if there is a path (of zero or more edges) from X to Y .

Answer:

```
reachable(X, X) :- node(X).
reachable(X, Y) :- edge(X, Z), reachable(Z, Y).
```

Note that we can't just use the clause $\text{reachable}(X, X)$, as that would not bind variable X in the body of clause, which violates the requirements of Datalog. That is, X is reachable from itself only if X is a node.

- (ii) Write a Datalog program that computes $\text{sameSCC}(X, Y)$, where $\text{sameSCC}(X, Y)$ holds if nodes X and node Y are in the same strongly connected component. (Hint: use the predicate reachable .)

Answer: *Two nodes a and b are in the same strongly connected component if and only if there is a path from a to b , and a path from b to a .*

```
sameSCC(X, Y) :- reachable(X, Y), reachable(Y, X).
```

For our example graph above, node a is in its own strongly connected component, but nodes b , c , and d are in the same SCC. So the result of the query $\text{sameSCC}(X, Y)$? is the following:

```
sameSCC(a, a)
sameSCC(b, b)
sameSCC(b, c)
sameSCC(b, d)
sameSCC(c, b)
sameSCC(c, c)
sameSCC(c, d)
sameSCC(d, b)
sameSCC(d, c)
sameSCC(d, d)
```